Examples of ML Functions Illustrating: Recursion, Patterns, Wildcard Variables, As-bindings, Let-environments, Options, and Basic I/O
(COP 4020/CIS 6930: Programming Languages)
(1) ML functions may be recursive.
fun r(L) =

  if L=nil

  then nil

  else r(tl(L)) @ [hd(L)];
What is r’s type?

What does r do?

(2) It’s often convenient to specify parameters with patterns.
fun r(nil) = nil

  | r(x::xs) = r(xs) @ [x];

What is r’s type?

(3) Let’s implement function r using difference lists.  One parameter keeps track of work remaining to be done, while another parameter keeps track of work already done.
fun rDiffLists(nil, processed) = processed

  | rDiffLists(x::xs, processed) = rDiffLists(xs, x::processed);

fun r(L) = rDiffLists(L, nil);

(4) More examples of patterns:
- fun f(3)=4
=   | f(n)=7;
val f = fn : int -> int

- f(5);

val it = 7 : int

- f(3);
val it = 4 : int

- fun f(3)=4;
stdIn:1.5-1.11 Warning: match nonexhaustive

          3 => ...  

val f = fn : int -> int

- f(3);
val it = 4 : int

- f(4);
uncaught exception Match [nonexhaustive match failure]...
(5) Wildcard, a.k.a. anonymous, variables can replace unused parameters, to unclutter code.

- fun f(3)=4
=   | f(_)=7;
val f = fn : int -> int

- f 4;
val it = 7 : int

(6) As-bindings can prevent having to reconstruct parameters.
fun inList(pair, nil) = false

  | inList(pair as (n,_), (n2,_)::L) = 
      if n=n2 then true else inList(pair,L);

- inList( (5,4), [(3,2),(1,0),(4,5)] );

val it = __________
- inList( (5,4), [(3,2),(1,0),(5,5)] );

val it = __________
(7) Functions can define local values (variables and functions) with let-environments.
  fun r(L) = 

  let

    fun rDiffLists(nil, processed) = processed

      | rDiffLists(x::xs, processed) = rDiffLists(xs, x::processed)

  in rDiffLists(L, nil)

  end;

(8) Another let-environment example, also illustrating static, versus dynamic, scope.
val v = 5;

fun f(x) =

  let

    fun g(x) = x+v

    fun h(x) =

      let val v = 3
      in g(v)

      end

    val v=6
    val _ = v+1

    fun pair(x) = (x,x)

    val (a,b) = pair(5)
  in

    h(v)
  end;
f(1);
(9) Another, more practical example:
  fun maxMiddle(L) =

  let

    fun findMax(n,nil) = n

      | findMax(n, (_,k,_)::L) = findMax(if k>n then k else n, L)

  in findMax(0,L)

  end;

- maxMiddle ([ (true,8,5), (true,12,12), (false,4,3) ]);

val it = ______
- maxMiddle [ (5,8,5.0), (5,12,4.3), (4,4,3.0) ];

val it = ______
(10) Options are a predefined data type in ML.  Options can either be empty of filled with some expression.  Values having type “T option” can either be NONE or SOME v (for a value v of type T).

- SOME(5);
val it = SOME 5 : int option
- NONE;

val it = NONE : 'a option
- SOME "hi";

val it = SOME "hi" : string option
- isSome(NONE);

val it = false : bool
- isSome(SOME 5);

val it = true : bool
- isSome;
val it = ____________________________________

- valOf(SOME 5);

val it = 5 : int
- valOf(NONE);

...uncaught exception Option...
- valOf;
val it = ____________________________________

(11) As with lists, patterns are convenient for analyzing option arguments.
fun sumList(L) =      

  let fun getSum(nil, sum) = sum

      | getSum(NONE::ns, sum) = getSum(ns, sum)

      | getSum(SOME(n)::ns, sum) = getSum(ns, n+sum)

  in getSum(L, 0)

  end;

sumList(NONE::SOME(4)::NONE::NONE::SOME(3)::SOME(2)::SOME(1)::[]);
(12) The only ML I/O we’ll use in this class is to print strings.

- print(if true then "hi" else "bye");
hival it = () : unit

- print;

val it = fn : string -> unit
(13) Expression sequences (e1;e2;…;en) are expressions that allow one subexpression to be executed after another.  The result of the expression sequence is the result of executing the last expression, en.  Expressions e1 to en-1 get evaluated just for their side effects (like I/O and memory updates using pointers, which we’ll discuss later in the semester).

- (print("hi"); "hi");
hival it = "hi" : string
(14) Exercise: Implement a function printAndAdd : int list->int, which prints all the elements of the argument list (separated by spaces) and then a newline, and returns the sum of the list elements.

